

```

; MDS-800 I/O DRIVERS FOR CP/M 2.2
; (FOUR DRIVE SINGLE DENSITY VERSION)
;
; VERSION 2.2 FEBRUARY, 1980
;
0016 = VERS EQU 22 ;VERSION 2.2
;
; COPYRIGHT (C) 1980
; DIGITAL RESEARCH
; BOX 579, PACIFIC GROVE
; CALIFORNIA, 93950
;
;
FFFF = TRUE EQU OFFFHH ;VALUE OF "TRUE"
0000 = FALSE EQU NOT TRUE ;"FALSE"
0000 = TEST EQU FALSE ;TRUE IF TEST BIOS
;
; IF TEST
BIAS EQU 03400H ;BASE OF CCF IN TEST SYSTEM
ENDIF
; IF NOT TEST
0000 = BIAS EQU 0000H ;GENERATE RELOCATABLE CP/M SYSTEM
ENDIF
;
1600 = PATCH EQU 1600H
;
1600 ORG PATCH
0000 = CPMB EQU $-PATCH ;BASE OF CPM CONSOLE PROCESSOR
0806 = BDOS EQU 806H+CPMB ;BASIC DOS (RESIDENT PORTION)
1600 = CPML EQU $-CPMB ;LENGTH (IN BYTES) OF CPM SYSTEM
002C = NSECTS EQU CPML/128 ;NUMBER OF SECTORS TO LOAD
0002 = OFFSET EQU 2 ;NUMBER OF DISK TRACKS USED BY CP/M
0004 = CDISK EQU 0004H ;ADDRESS OF LAST LOGGED DISK ON WARM START
0080 = BUFF EQU 0080H ;DEFAULT BUFFER ADDRESS
000A = RETRY EQU 10 ;MAX RETRIES ON DISK I/O BEFORE ERROR
;
; PERFORM FOLLOWING FUNCTIONS
; BOOT COLD START
; WBOOT WARM START (SAVE I/O BYTE)
; (BOOT AND WBOOT ARE THE SAME FOR MDS)
; CONST CONSOLE STATUS
; REG-A = 00 IF NO CHARACTER READY
; REG-A = FF IF CHARACTER READY
; CONIN CONSOLE CHARACTER IN (RESULT IN REG-A)
; CONOUT CONSOLE CHARACTER OUT (CHAR IN REG-C)
; LIST LIST OUT (CHAR IN REG-C)
; PUNCH PUNCH OUT (CHAR IN REG-C)
; READER PAPER TAPE READER IN (RESULT TO REG-A)
; HOME MOVE TO TRACK 00
;
; (THE FOLLOWING CALLS SET-UP THE IO PARAMETER BLOCK FOR THE
; MDS, WHICH IS USED TO PERFORM SUBSEQUENT READS AND WRITES)
; SELDSK SELECT DISK GIVEN BY REG-C (0,1,2,...)
; SETTRK SET TRACK ADDRESS (0,...76) FOR SUBSEQUENT READ/WRITE
; SETSEC SET SECTOR ADDRESS (1,...,26) FOR SUBSEQUENT READ/WRITE
; SETDMA SET SUBSEQUENT DMA ADDRESS (INITIALLY 80H)

```

CP/M 2.2	
COPYRIGHT ©	1980
DIGITAL RESEARCH, INC.	
P. O. BOX 579	
PACIFIC GROVE, CA 93950	
SERIAL #	L-756

```

;
; (READ AND WRITE ASSUME PREVIOUS CALLS TO SET UP THE IO PARAMETER
; READ READ TRACK/SECTOR TO PRESET DMA ADDRESS
; WRITE WRITE TRACK/SECTOR FROM PRESET DMA ADDRESS
;

```

JUMP VECTOR FOR INDIVIDUAL ROUTINES

```

1600 C3B316 JMP BOOT
1603 C3C316 WBOOTE: JMP WBOOT
1606 C36117 JMP CONST
1609 C36417 JMP CONIN
160C C36A17 JMP CONOUT
160F C36D17 JMP LIST
1612 C37217 JMP PUNCH
1615 C37517 JMP READER
1618 C37817 JMP HOME
161B C37D17 JMP SELDSK
161E C3A717 JMP SETTRK
1621 C3AC17 JMP SETSEC
1624 C3BB17 JMP SETDMA
1627 C3C117 JMP READ
162A C3CA17 JMP WRITE
162D C37017 JMP LISTST ;LIST STATUS
1630 C3B117 JMP SECTAN

```


COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950
 SERIAL # _____

```

;
; MACLIB DISKDEF ;LOAD THE DISK DEFINITION LIBRARY
; DISKS 4 ;FOUR DISKS
1633+= DPBASE EQU $ ;BASE OF DISK PARAMETER BLOCKS
1633+82160000 DPE0: DW XLTO,0000H ;TRANSLATE TABLE
1637+00000000 DW 0000H,0000H ;SCRATCH AREA
163B+6E187316 DW DIRBUF,DPB0 ;DIR BUFF,PARM BLOCK
163F+0D19EE18 DW CSV0,ALV0 ;CHECK, ALLOC VECTORS
1643+82160000 DPE1: DW XLT1,0000H ;TRANSLATE TABLE
1647+00000000 DW 0000H,0000H ;SCRATCH AREA
164B+6E187316 DW DIRBUF,DPB1 ;DIR BUFF,PARM BLOCK
164F+3C191D19 DW CSV1,ALV1 ;CHECK, ALLOC VECTORS
1653+82160000 DPE2: DW XLT2,0000H ;TRANSLATE TABLE
1657+00000000 DW 0000H,0000H ;SCRATCH AREA
165B+6E187316 DW DIRBUF,DPB2 ;DIR BUFF,PARM BLOCK
165F+6B194C19 DW CSV2,ALV2 ;CHECK, ALLOC VECTORS
1663+82160000 DPE3: DW XLT3,0000H ;TRANSLATE TABLE
1667+00000000 DW 0000H,0000H ;SCRATCH AREA
166B+6E187316 DW DIRBUF,DPB3 ;DIR BUFF,PARM BLOCK
166F+9A197B19 DW CSV3,ALV3 ;CHECK, ALLOC VECTORS
DISKDEF 0,1,26,6,1024,243,64,64,OFFSET
1673+= DPB0 EQU $ ;DISK PARM BLOCK
1673+1A00 DW 26 ;SEC PER TRACK
1675+03 DB 3 ;BLOCK SHIFT
1676+07 DB 7 ;BLOCK MASK
1677+00 DB 0 ;EXTNT MASK
1678+F200 DW 242 ;DISK SIZE-1
167A+3F00 DW 63 ;DIRECTORY MAX
167C+C0 DB 192 ;ALLOCO
167D+00 DB 0 ;ALLOC1
167E+1000 DW 16 ;CHECK SIZE
1680+0200 DW 2 ;OFFSET
1682+= XLTO EQU $ ;TRANSLATE TABLE

```

1682+01	DB	1
1683+07	DB	7
1684+0D	DB	13
1685+13	DB	19
1686+19	DB	25
1687+05	DB	5
1688+0B	DB	11
1689+11	DB	17
168A+17	DB	23
168B+03	DB	3
168C+09	DB	9
168D+0F	DB	15
168E+15	DB	21
168F+02	DB	2
1690+08	DB	8
1691+0E	DB	14
1692+14	DB	20
1693+1A	DB	26
1694+06	DB	6
1695+0C	DB	12
1696+12	DB	18
1697+18	DB	24
1698+04	DB	4
1699+0A	DB	10
169A+10	DB	16
169B+16	DB	22



1673+=	DPB1	EQU	DPB0	;EQUIVALENT PARAMETERS
001F+=	ALS1	EQU	ALSO	;SAME ALLOCATION VECTOR SIZE
0010+=	CSS1	EQU	CSS0	;SAME CHECKSUM VECTOR SIZE
1682+=	XLT1	EQU	XLT0	;SAME TRANSLATE TABLE

DISKDEF 2,0

1673+=	DPB2	EQU	DPB0	;EQUIVALENT PARAMETERS
001F+=	ALS2	EQU	ALSO	;SAME ALLOCATION VECTOR SIZE
0010+=	CSS2	EQU	CSS0	;SAME CHECKSUM VECTOR SIZE
1682+=	XLT2	EQU	XLT0	;SAME TRANSLATE TABLE

DISKDEF 3,0

1673+=	DPB3	EQU	DPB0	;EQUIVALENT PARAMETERS
001F+=	ALS3	EQU	ALSO	;SAME ALLOCATION VECTOR SIZE
0010+=	CSS3	EQU	CSS0	;SAME CHECKSUM VECTOR SIZE
1682+=	XLT3	EQU	XLT0	;SAME TRANSLATE TABLE

; ENDEF OCCURS AT END OF ASSEMBLY

;
;
;
;
;
;
;
;
;
;
;
;

END OF CONTROLLER - INDEPENDENT CODE, THE REMAINING SUBROUTINES ARE TAILORED TO THE PARTICULAR OPERATING ENVIRONMENT, AND MUST BE ALTERED FOR ANY SYSTEM WHICH DIFFERS FROM THE INTEL MDS.

THE FOLLOWING CODE ASSUMES THE MDS MONITOR EXISTS AT OF800H AND USES THE I/O SUBROUTINES WITHIN THE MONITOR

WE ALSO ASSUME THE MDS SYSTEM HAS FOUR DISK DRIVES

00FD =	REVRT	EQU	OFDH	;INTERRUPT REVERT PORT
00FC =	INTC	EQU	OFCH	;INTERRUPT MASK PORT
00F3 =	ICON	EQU	OF3H	;INTERRUPT CONTROL PORT
007E =	INTE	EQU	0111\$1110B	;ENABLE RST 0(WARM BOOT), RST 7 (MONITOR)

;

```

; MDS MONITOR EQUATES
F800 = MON80 EQU 0F800H ;MDS MONITOR
FF0F = RMON80 EQU 0FF0FH ;RESTART MON80 (BOOT ERROR)
F803 = CI EQU 0F803H ;CONSOLE CHARACTER TO REG-A
F806 = RI EQU 0F806H ;READER IN TO REG-A
F809 = CO EQU 0F809H ;CONSOLE CHAR FROM C TO CONSOLE OUT
F80C = PO EQU 0F80CH ;PUNCH CHAR FROM C TO PUNCH DEVICE
F80F = LO EQU 0F80FH ;LIST FROM C TO LIST DEVICE
F812 = CSTS EQU 0F812H ;CONSOLE STATUS 00/FF TO REGISTER A

```

```

; DISK PORTS AND COMMANDS
0078 = BASE EQU 78H ;BASE OF DISK COMMAND IO PORTS
0078 = DSTAT EQU BASE ;DISK STATUS (INPUT)
0079 = RTYPE EQU BASE+1 ;RESULT TYPE (INPUT)
007B = RBYTE EQU BASE+3 ;RESULT BYTE (INPUT)

```

```

;
0079 = ILOW EQU BASE+1 ;IOPB LOW ADDRESS (OUTPUT)
007A = IHIGH EQU BASE+2 ;IOPB HIGH ADDRESS (OUTPUT)

```

```

;
0004 = READF EQU 4H ;READ FUNCTION
0006 = WRITF EQU 6H ;WRITE FUNCTION
0003 = RECAL EQU 3H ;RECALIBRATE DRIVE
0004 = IORDY EQU 4H ;I/O FINISHED MASK
000D = CR EQU 0DH ;CARRIAGE RETURN
000A = LF EQU 0AH ;LINE FEED

```

```

;
SIGNON: ;SIGNON MESSAGE: XXK CP/M VERS Y.Y
169C 0D0A0A DB CR,LF,LF
IF TEST
DB '32' ;32K EXAMPLE BIOS
ENDIF
IF NOT TEST
169F 3030 DB '00' ;MEMORY SIZE FILLED BY RELOCATOR
ENDIF
16A1 6B2043502F DB 'k CP/M vers '
16AD 322E32 DB VERS/10+'0','.',VERS MOD 10+'0'
16B0 0D0A00 DB CR,LF,0

```

```

;
BOOT: ;PRINT SIGNON MESSAGE AND GO TO CCP
; (NOTE: MDS BOOT INITIALIZED IOBYTE AT 0003H)
16B3 310001 LXI SP,BUFF+80H
16B6 219C16 LXI H,SIGNON
16B9 CDD317 CALL PRMSG ;PRINT MESSAGE
16BC AF XRA A ;CLEAR ACCUMULATOR
16BD 320400 STA CDISK ;SET INITIALLY TO DISK A
16C0 C30F17 JMP GOCPM ;GO TO CP/M

```

```

;
WBOOT:; LOADER ON TRACK 0, SECTOR 1, WHICH WILL BE SKIPPED FOR WARM
; READ CP/M FROM DISK - ASSUMING THERE IS A 128 BYTE COLD START
; START.
;

```

```

16C3 318000 LXI SP,BUFF ;USING DMA - THUS 80 THRU FF AVAILABLE FOR STACK
;
16C6 0E0A MVI C,RETRY ;MAX RETRIES
16C8 C5 PUSH B

```

COPYRIGHT © 1980

DIGITAL RESEARCH, INC.

P. O. BOX 579

FARMING GROVE, CA 93950

SERIAL #

```

WBOOT0: ;ENTER HERE ON ERROR RETRIES
16C9 010000 LXI B,CMPB ;SET DMA ADDRESS TO START OF DISK SYSTEM
16CC CDBB17 CALL SETDMA
16CF 0E00 MVI C,0 ;BOOT FROM DRIVE 0
16D1 CD7D17 CALL SELDSK
16D4 0E00 MVI C,0
16D6 CDA717 CALL SETTRK ;START WITH TRACK 0
16D9 0E02 MVI C,2 ;START READING SECTOR 2
16DB CDAC17 CALL SETSEC

;
; READ SECTORS, COUNT NSECTS TO ZERO
16DE C1 POP B ;10-ERROR COUNT
16DF 062C MVI B,NSECTS
RDSEC: ;READ NEXT SECTOR
16E1 C5 PUSH B ;SAVE SECTOR COUNT
16E2 CDC117 CALL READ
16E5 C24917 JNZ BOOTERR ;RETRY IF ERRORS OCCUR
16E8 2A6C18 LHLD IOD ;INCREMENT DMA ADDRESS
16EB 118000 LXI D,128 ;SECTOR SIZE
16EE 19 DAD D ;INCREMENTED DMA ADDRESS IN HL
16EF 44 MOV B,H
16F0 4D MOV C,L ;READY FOR CALL TO SET DMA
16F1 CDBB17 CALL SETDMA
16F4 3A6B18 LDA IOS ;SECTOR NUMBER JUST READ
16F7 FE1A CPI 26 ;READ LAST SECTOR?
16F9 DA0517 JC RD1

; MUST BE SECTOR 26, ZERO AND GO TO NEXT TRACK
16FC 3A6A18 LDA IOT ;GET TRACK TO REGISTER A
16FF 3C INR A
1700 4F MOV C,A ;READY FOR CALL
1701 CDA717 CALL SETTRK
1704 AF XRA A ;CLEAR SECTOR NUMBER
1705 3C RD1: INR A ;TO NEXT SECTOR
1706 4F MOV C,A ;READY FOR CALL
1707 CDAC17 CALL SETSEC
170A C1 POP B ;RECALL SECTOR COUNT
170B 05 DCR B ;DONE?
170C C2E116 JNZ RDSEC

;
; DONE WITH THE LOAD, RESET DEFAULT BUFFER ADDRESS
GOCPM: ;(ENTER HERE FROM COLD START BOOT)
; ENABLE RST0 AND RST7
170F F3 DI
1710 3E12 MVI A,12H ;INITIALIZE COMMAND
1712 D3FD OUT REVRT
1714 AF XRA A
1715 D3FC OUT INTC ;CLEARED
1717 3E7E MVI A,INTE ;RST0 AND RST7 BITS ON
1719 D3FC OUT INTC
171B AF XRA A
171C D3F3 OUT ICON ;INTERRUPT CONTROL

;
; SET DEFAULT BUFFER ADDRESS TO 80H
171E 018000 LXI B,BUFF
1721 CDBB17 CALL SETDMA

```

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 FORT WORTH, TX 76101

SURE #

```

; RESET MONITOR ENTRY POINTS
1724 3EC3      MVI      A,JMP
1726 320000    STA      0
1729 210316    LXI      H,WBOOT
172C 220100    SHLD     1      ;JMP WBOOT AT LOCATION 00
172F 320500    STA      5
1732 210608    LXI      H,BDOS
1735 220600    SHLD     6      ;JMP BDOS AT LOCATION 5
; IF NOT TEST
1738 323800    STA      7*8    ;JMP TO MON80 (MAY HAVE BEEN CHANGED BY DDT)
173B 2100F8    LXI      H,MON80
173E 223900    SHLD     7*8+1
; ENDIF
; LEAVE IOBYTE SET
; PREVIOUSLY SELECTED DISK WAS B, SEND PARAMETER TO CPM
1741 3A0400    LDA      CDISK   ;LAST LOGGED DISK NUMBER
1744 4F        MOV      C,A     ;SEND TO CCP TO LOG IT IN
1745 FB        EI
1746 C30000    JMP      CPMB
;
; ERROR CONDITION OCCURRED, PRINT MESSAGE AND RETRY
BOOTERR:
1749 C1        POP      B       ;RECALL COUNTS
174A 0D        DCR      C
174B CA5217    JZ       BOOTERO
; TRY AGAIN
174E C5        PUSH     B
174F C3C916    JMP      WBOOT0
;
BOOTERO:
; OTHERWISE TOO MANY RETRIES
1752 215B17    LXI      H,BOOTMSG
1755 CDD317    CALL     PRMSG
1758 C30FFF    JMP      RMON80  ;MDS HARDWARE MONITOR
;
BOOTMSG:
175B 3F626F6F74 DB      '?boot',0
;
;
CONST: ;CONSOLE STATUS TO REG-A
; (EXACTLY THE SAME AS MDS CALL)
1761 C312F8    JMP      CSTS
;
CONIN: ;CONSOLE CHARACTER TO REG-A
1764 CD03F8    CALL     CI
1767 E67F     ANI      7FH     ;REMOVE PARITY BIT
1769 C9        RET
;
CONOUT: ;CONSOLE CHARACTER FROM C TO CONSOLE OUT
176A C309F8    JMP      CO
;
LIST: ;LIST DEVICE OUT
; (EXACTLY THE SAME AS MDS CALL)
176D C30FF8    JMP      LO
;
LISTST:

```

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950
 SERIAL # _____

```

;RETURN LIST STATUS
1770 AF      XRA      A
1771 C9      RET              ;ALWAYS NOT READY

;
PUNCH:      ;PUNCH DEVICE OUT
;            (EXACTLY THE SAME AS MDS CALL)
1772 C30CF8  JMP      PO

;
READER:     ;READER CHARACTER IN TO REG-A
;            (EXACTLY THE SAME AS MDS CALL)
1775 C306F8  JMP      RI

;
HOME:       ;MOVE TO HOME POSITION
;            TREAT AS TRACK 00 SEEK
1778 0E00    MVI      C,0
177A C3A717  JMP      SETTRK

;
SELDSK:     ;SELECT DISK GIVEN BY REGISTER C
177D 210000  LXI      H,000H ;RETURN 0000 IF ERROR
1780 79      MOV      A,C
1781 FE04    CPI      NDISKS ;TOO LARGE?
1783 D0      RNC              ;LEAVE HL = 0000

;
1784 E602    ANI      10B      ;00 00 FOR DRIVE 0,1 AND 10 10 FOR DRIVE 2,3
1786 326618  STA      DBANK    ;TO SELECT DRIVE BANK
1789 79      MOV      A,C      ;00, 01, 10, 11
178A E601    ANI      1B       ;MDS HAS 0,1 AT 78, 2,3 AT 88
178C B7      ORA      A        ;RESULT 00?
178D CA9217  JZ        SETDRIVE
1790 3E30    MVI      A,00110000B ;SELECTS DRIVE 1 IN BANK

SETDRIVE:
1792 47      MOV      B,A      ;SAVE THE FUNCTION
1793 216818  LXI      H,IOF    ;IO FUNCTION
1796 7E      MOV      A,M
1797 E6CF    ANI      11001111B ;MASK OUT DISK NUMBER
1799 B0      ORA      B        ;MASK IN NEW DISK NUMBER
179A 77      MOV      M,A      ;SAVE IT IN IOPB
179B 69      MOV      L,C
179C 2600    MVI      H,0      ;HL=DISK NUMBER
179E 29      DAD      H        ;*2
179F 29      DAD      H        ;*4
17A0 29      DAD      H        ;*8
17A1 29      DAD      H        ;*16
17A2 113316  LXI      D,DPBASE
17A5 19      DAD      D        ;HL=DISK HEADER TABLE ADDRESS
17A6 C9      RET

;
;
SETTRK:     ;SET TRACK ADDRESS GIVEN BY C
17A7 216A18  LXI      H,IOT
17AA 71      MOV      M,C
17AB C9      RET

;
SETSEC:     ;SET SECTOR NUMBER GIVEN BY C
17AC 216B18  LXI      H,IOS
17AF 71      MOV      M,C

```

COPYRIGHT © 1980	
DIGITAL RESEARCH, INC.	
P. O. BOX 570	
PACIFIC GROVE, CA 93950	
SERIAL #	_____

```

17B0 C9          RET
                SECTRAN:
                                ;TRANSLATE SECTOR BC USING TABLE AT DE
17B1 0600        MVI      B,0      ;DOUBLE PRECISION SECTOR NUMBER IN BC
17B3 EB          XCHG      ;TRANSLATE TABLE ADDRESS TO HL
17B4 09          DAD       B        ;TRANSLATE(SECTOR) ADDRESS
17B5 7E          MOV       A,M      ;TRANSLATED SECTOR NUMBER TO A
17B6 326B18      STA       IOS
17B9 6F          MOV       L,A      ;RETURN SECTOR NUMBER IN L
17BA C9          RET

;
SETDMA: ;SET DMA ADDRESS GIVEN BY REGS B,C
17BB 69          MOV       L,C
17BC 60          MOV       H,B
17BD 226C18      SHLD      IOD
17C0 C9          RET

;
READ: ;READ NEXT DISK RECORD (ASSUMING DISK/TRK/SEC/DMA SET)
17C1 0E04        MVI       C,READF ;SET TO READ FUNCTION
17C3 CDE017      CALL      SETFUNC
17C6 CDF017      CALL      WAITIO ;PERFORM READ FUNCTION
17C9 C9          RET              ;MAY HAVE ERROR SET IN REG-A

;
;
WRITE: ;DISK WRITE FUNCTION
17CA 0E06        MVI       C,WRITF
17CC CDE017      CALL      SETFUNC ;SET TO WRITE FUNCTION
17CF CDF017      CALL      WAITIO
17D2 C9          RET              ;MAY HAVE ERROR SET

;
;
;
UTILITY SUBROUTINES
PRMSG: ;PRINT MESSAGE AT H,L TO 0
17D3 7E          MOV       A,M
17D4 B7          ORA       A        ;ZERO?
17D5 C8          RZ

;
;
;
MORE TO PRINT
17D6 E5          PUSH      H
17D7 4F          MOV       C,A
17D8 CD6A17      CALL      CONOUT
17DB E1          POP       H
17DC 23          INX       H
17DD C3D317      JMP       PRMSG

;
SETFUNC:
;
SET FUNCTION FOR NEXT I/O (COMMAND IN REG-C)
17E0 216818      LXI       H,IOF   ;IO FUNCTION ADDRESS
17E3 7E          MOV       A,M      ;GET IT TO ACCUMULATOR FOR MASKING
17E4 E6F8        ANI       11111000B ;REMOVE PREVIOUS COMMAND
17E6 B1          ORA       C        ;SET TO NEW COMMAND
17E7 77          MOV       M,A      ;REPLACED IN IOPB

;
;
THE MDS-800 CONTROLLER REQUIRES DISK BANK BIT IN SECTOR BYTE
MASK THE BIT FROM THE CURRENT I/O FUNCTION
17E8 E620        ANI       00100000B ;MASK THE DISK SELECT BIT
17EA 216B18      LXI       H,IOS    ;ADDRESS THE SECTOR SELECT BYTE
17ED B6          ORA       M        ;SELECT PROPER DISK BANK

```

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950
 SERIAL # _____


```

17EE 77      MOV      M,A          ;SET DISK SELECT BIT ON/OFF
17EF C9      RET

;
WAITIO:
17F0 0EOA    MVI      C,RETRY ;MAX RETRIES BEFORE PERM ERROR
REWAIT:
;          START THE I/O FUNCTION AND WAIT FOR COMPLETION
17F2 CD3F18  CALL     INTYPE ;IN RTYPE
17F5 CD4C18  CALL     INBYTE ;CLEARS THE CONTROLLER
;
17F8 3A6618  LDA      DBANK        ;SET BANK FLAGS
17FB B7      ORA      A            ;ZERO IF DRIVE 0,1 AND NZ IF 2,3
17FC 3E67    MVI      A,IOPB AND OFFH ;LOW ADDRESS FOR IOPB
17FE 0618    MVI      B,IOPB SHR 8  ;HIGH ADDRESS FOR IOPB
1800 C20B18  JNZ      IODR1 ;DRIVE BANK 1?
1803 D379    OUT      ILOW         ;LOW ADDRESS TO CONTROLLER
1805 78      MOV      A,B
1806 D37A    OUT      IHIGH        ;HIGH ADDRESS
1808 C31018  JMP      WAITO        ;TO WAIT FOR COMPLETE
;
IODR1: ;DRIVE BANK 1
180B D389    OUT      ILOW+10H      ;88 FOR DRIVE BANK 10
180D 78      MOV      A,B
180E D38A    OUT      IHIGH+10H
;
1810 CD5918  WAITO:  CALL     INSTAT ;WAIT FOR COMPLETION
1813 E604    ANI      IORDY        ;READY?
1815 CA1018  JZ       WAITO
;
;          CHECK IO COMPLETION OK
1818 CD3F18  CALL     INTYPE        ;MUST BE IO COMPLETE (00) UNLINKED
;          00 UNLINKED I/O COMPLETE, 01 LINKED I/O COMPLETE (NOT USED)
;          10 DISK STATUS CHANGED 11 (NOT USED)
181B FE02    CPI      10B          ;READY STATUS CHANGE?
181D CA3218  JZ       WREADY
;
;          MUST BE 00 IN THE ACCUMULATOR
1820 B7      ORA      A
1821 C23818  JNZ      WERROR        ;SOME OTHER CONDITION, RETRY
;
;          CHECK I/O ERROR BITS
1824 CD4C18  CALL     INBYTE
1827 17      RAL
1828 DA3218  JC       WREADY        ;UNIT NOT READY
182B 1F      RAR
182C E6FE    ANI      11111110B    ;ANY OTHER ERRORS? (DELETED DATA OK)
182E C23818  JNZ      WERROR
;
;          READ OR WRITE IS OK, ACCUMULATOR CONTAINS ZERO
1831 C9      RET
;
WREADY: ;NOT READY, TREAT AS ERROR FOR NOW
1832 CD4C18  CALL     INBYTE        ;CLEAR RESULT BYTE
1835 C33818  JMP      TRYCOUNT
;
WERROR: ;RETURN HARDWARE MALFUNCTION (CRC, TRACK, SEEK, ETC.)

```

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950

STRIP: #

```

; THE MDS CONTROLLER HAS RETURNED A BIT IN EACH POSITION
; OF THE ACCUMULATOR, CORRESPONDING TO THE CONDITIONS:
; 0 - DELETED DATA (ACCEPTED AS OK ABOVE)
; 1 - CRC ERROR
; 2 - SEEK ERROR
; 3 - ADDRESS ERROR (HARDWARE MALFUNCTION)
; 4 - DATA OVER/UNDER FLOW (HARDWARE MALFUNCTION)
; 5 - WRITE PROTECT (TREATED AS NOT READY)
; 6 - WRITE ERROR (HARDWARE MALFUNCTION)
; 7 - NOT READY
; (ACCUMULATOR BITS ARE NUMBERED 7 6 5 4 3 2 1 0)
;
; IT MAY BE USEFUL TO FILTER OUT THE VARIOUS CONDITIONS,
; BUT WE WILL GET A PERMANENT ERROR MESSAGE IF IT IS NOT
; RECOVERABLE. IN ANY CASE, THE NOT READY CONDITION IS
; TREATED AS A SEPARATE CONDITION FOR LATER IMPROVEMENT
TRYCOUNT:
; REGISTER C CONTAINS RETRY COUNT, DECREMENT 'TIL ZERO
1838 OD DCR C
1839 C2F217 JNZ REWAIT ;FOR ANOTHER TRY
;
; CANNOT RECOVER FROM ERROR
183C 3E01 MVI A,1 ;ERROR CODE
183E C9 RET
;
; INTYPE, INBYTE, INSTAT READ DRIVE BANK 00 OR 10
183F 3A6618 INTYPE: LDA DBANK
1842 B7 ORA A
1843 C24918 JNZ INTYP1 ;SKIP TO BANK 10
1846 DB79 IN RTYPE
1848 C9 RET
1849 DB89 INTYP1: IN RTYPE+10H ;78 FOR 0,1 88 FOR 2,3
184B C9 RET
;
; INBYTE: LDA DBANK
184C 3A6618 INBYTE: LDA DBANK
184F B7 ORA A
1850 C25618 JNZ INBYT1
1853 DB7B IN RBYTE
1855 C9 RET
1856 DB8B INBYT1: IN RBYTE+10H
1858 C9 RET
;
; INSTAT: LDA DBANK
1859 3A6618 INSTAT: LDA DBANK
185C B7 ORA A
185D C26318 JNZ INSTA1
1860 DB78 IN DSTAT
1862 C9 RET
1863 DB88 INSTA1: IN DSTAT+10H
1865 C9 RET
;
; DATA AREAS (MUST BE IN RAM)
1866 00 DBANK: DB 0 ;DISK BANK 00 IF DRIVE 0,1
; 10 IF DRIVE 2,3
IOPB: ;IO PARAMETER BLOCK

```

COPYRIGHT © 1980

DIGITAL RESEARCH, INC.

P. O. BOX 579
PACIFIC GROVE, CA 93950

SERIAL # _____

```

1867 80          DB      80H      ;NORMAL I/O OPERATION
1868 04          IOF:    DB      READF ;IO FUNCTION, INITIAL READ
1869 01          ION:    DB      1      ;NUMBER OF SECTORS TO READ
186A 02          IOT:    DB      OFFSET ;TRACK NUMBER
186B 01          IOS:    DB      1      ;SECTOR NUMBER
186C 8000        IOD:    DW      BUFF   ;IO ADDRESS

```

```

;
;
;

```

```

        DEFINE RAM AREAS FOR BDOS OPERATION

```

```

        ENDEF

```

```

186E+=          BEGDAT EQU      $
186E+          DIRBUF: DS      128      ;DIRECTORY ACCESS BUFFER
18EE+          ALV0:   DS      31
190D+          CSV0:   DS      16
191D+          ALV1:   DS      31
193C+          CSV1:   DS      16
194C+          ALV2:   DS      31
196B+          CSV2:   DS      16
197B+          ALV3:   DS      31
199A+          CSV3:   DS      16
19AA+=          ENDDAT EQU      $
013C+=          DATSIZ EQU      $-BEGDAT
19AA          END

```

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950

SERIAL # _____

001F ALS1	001F ALS2	001F ALS3	18EE ALVO	191D ALV1
194C ALV2	197B ALV3	0078 BASE	0806 BDOS	186E BEGDAT
0000 BIAS	16B3 BOOT	1752 BOOTERO	1749 BOOTERR	175B BOOTMSG
0080 BUFF	0004 CDISK	F803 CI	1764 CONIN	176A CONOUT
1761 CONST	F809 CO	0000 CPMB	1600 CPML	000D CR
0010 CSS1	0010 CSS2	0010 CSS3	F812 CSTS	190D CSVO
193C CSV1	196B CSV2	199A CSV3	013C DATSIZ	1866 DBANK
186E DIRBUF	1673 DPB0	1673 DPB1	1673 DPB2	1673 DPB3
1633 DPBASE	1633 DPE0	1643 DPE1	1653 DPE2	1663 DPE3
0078 DSTAT	19AA ENDDAT	0000 FALSE	170F GOCPM	1778 HOME
00F3 ICON	007A IHIGH	0079 ILOW	1856 INBYT1	184C INBYTE
1863 INSTA1	1859 INSTAT	00FC INTC	007E INTE	1849 INTYP1
33F INTYPE	186C IOD	180B IO'R1	1868 IOF	1869 ION
1867 IOPB	0004 IORDY	186B IOS	186A IOT	000A LF
176D LIST	1770 LISTST	F80F LO	F800 MON80	002C NSECTS
0002 OFFSET	1600 PATCH	F80C PO	17D3 PRMSG	1772 PUNCH
007B RBYTE	1705 RD1	16E1 RDSEC	1775 READER	17C1 READ
0004 READF	0003 RECAL	000A RETRY	00FD REVRT	17F2 REWAIT
F306 RI	FF0F RMON80	0079 RTYPE	17B1 SECTRAN	177D SELDSK
17BB SETDMA	1792 SETDRIVE	17E0 SEIFUNC	17AC SETSEC	17A7 SETTRK
169C SIGNON	0000 TEST	FFFF TRUE	1838 TRYCOUNT	0016 VERS
1810 WAITO	17F0 WAITIO	16C9 WBOOTO	1603 WBOOTE	16C3 WBOOT
1838 WERROR	1832 WREADY	17CA WRITE	0006 WRITF	1682 XLTO
1682 XLT1	1682 XLT2	1682 XLT3		

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950

SERIAL # _____